

# Examples in Maple

November 14, 2023 10:22 AM

Let  $\omega$  be a 5th root of unity. Solve  $(1 - \omega) \cdot y = \omega^4 - \omega^2$  for  $y$  by computing the inverse of  $(1 - \omega)$  via the inverse of  $(1 - z)$  in  $\mathbb{Q}[z]/m(z)$  where  $m(z) = z^4 + z^3 + z^2 + z + 1$ .

```
> m := z^4+z^3+z^2+z+1;
```

$$m := z^4 + z^3 + z^2 + z + 1 \quad (1)$$

```
> gcdex(1-z,m,z,'s');
```

$$1 \quad (2)$$

```
> s;
```

$$\frac{1}{5} z^3 + \frac{2}{5} z^2 + \frac{3}{5} z + \frac{4}{5} \quad (3)$$

```
> y = subs(z=omega, rem( s*(z^4+z^2), m, z ) );
```

$$y = -\frac{3}{5} \omega^3 - \frac{1}{5} \omega^2 - \frac{4}{5} \omega - \frac{2}{5} \quad (4)$$

Maple's RootOf representation for algebraic numbers

```
> omega := RootOf(m,z);
```

$$\omega := \text{RootOf}(\_Z^4 + \_Z^3 + \_Z^2 + \_Z + 1) \quad (5)$$

```
> evala( omega^5 );
```

$$1 \quad (6)$$

```
> evala( 1/(1-omega) );
```

$$\frac{\text{RootOf}(\_Z^4 + \_Z^3 + \_Z^2 + \_Z + 1)^3}{5} + \frac{2 \text{RootOf}(\_Z^4 + \_Z^3 + \_Z^2 + \_Z + 1)^2}{5} + \frac{3 \text{RootOf}(\_Z^4 + \_Z^3 + \_Z^2 + \_Z + 1)}{5} + \frac{4}{5} \quad (7)$$

```
> omega := 'omega': alias(omega=RootOf(m,z));
```

```
> evala(omega^6);
```

$$\omega \quad (8)$$

```
> evala(1/(1-omega));
```

$$\frac{1}{5} \omega^3 + \frac{2}{5} \omega^2 + \frac{3}{5} \omega + \frac{4}{5} \quad (9)$$

```
> solve( {omega*x+omega*y=1,omega^3*x+omega^4*y=-1}, {x,y} );
```

$$\left\{ x = -\frac{2}{5} \omega^3 - \frac{4}{5} \omega^2 - \frac{1}{5} \omega - \frac{3}{5}, y = -\frac{3}{5} \omega^3 - \frac{1}{5} \omega^2 - \frac{4}{5} \omega - \frac{2}{5} \right\} \quad (10)$$

```
> convert(omega,radical);
```

$$\frac{\sqrt{5}}{4} - \frac{1}{4} + \frac{I\sqrt{2}\sqrt{5+\sqrt{5}}}{4} \quad (11)$$

```
> evalf(omega);
```

$$0.3090169944 + 0.9510565163 I \quad (12)$$

The Cyclotomic polynomials

```
> with(numtheory):
```

The cyclotomic polynomials

**> with(numtheory):**

**> cyclotomic(5,z);**

$$z^4 + z^3 + z^2 + z + 1 \quad (13)$$

**> seq( cyclotomic(n,z), n=1..6 );**

$$z - 1, z + 1, z^2 + z + 1, z^2 + 1, z^4 + z^3 + z^2 + z + 1, z^2 - z + 1 \quad (14)$$

Minimal polynomials

**> with(PolynomialTools):**

**> MinimalPolynomial(omega,z);**

$$z^4 + z^3 + z^2 + z + 1 \quad (15)$$

**> alpha := 1+sqrt(2)+sqrt(3);**

$$\alpha := 1 + \sqrt{2} + \sqrt{3} \quad (16)$$

**> MinimalPolynomial(alpha,z);**

$$z^4 - 4z^3 - 4z^2 + 16z - 8 \quad (17)$$

Factor  $m(z)$  over  $\mathbb{Q}$

**> factor(m);**

$$z^4 + z^3 + z^2 + z + 1 \quad (18)$$

Factor  $m(z)$  over  $\mathbb{Q}(\omega)$

**> factor(m,omega);**

$$-(\omega^3 + \omega^2 + \omega + \underline{z} + 1) (\omega^2 - \underline{z}) (\omega^3 - \underline{z}) (\underline{-z} + \omega) \quad (19)$$

